Staff Papers Series

Staff Paper P86-29 September 1986

AN APPLICATION OF BAYESIAN VECTOR AUTOREGRESSION

TO THE U.S. TURKEY MARKET

by

Steve Ford

Department of Agricultural and Applied Economics

University of Minnesota
Institute of Agriculture, Forestry and Home Economics
St. Paul, Minnesota 55108

Staff Paper P86-29 September 1986

AN APPLICATION OF BAYESIAN VECTOR AUTOREGRESSION TO THE U.S. TURKEY MARKET

by

Steve Ford

I. INTRODUCTION

The purpose of this paper is to illustrate how the specification of Bayesian priors in a vector autoregression (VAR) can dramatically alter the results of the estimation of such a model. A new approach is used in this paper to modify the traditional use of Bayesian priors in VAR's to fit the needs of a highly seasonal model of the U.S. turkey market.

A prior knowledge of VAR methodology is assumed in this paper. Ford [3] presents a cursory discussion of vector autoregression methodology and provides references to other articles in the literature. Section II of this paper discusses the mechanics involved in setting the Bayesian priors and their use in the estimation technique. Section III compares two VAR's of the U.S. turkey market; one model with standard Bayesian priors and one model with the priors adjusted to account for seasonality in the market. A list defining variables used in this paper and the RATS[1] code used in the estimation of the model presented in this paper appear in the appendix.

---

[1] Regression Analysis of Time Series by Thomas Doan and Robert Litterman, VAR Econometrics, Minneapolis, Minnesota.

II. BAYESIAN VECTOR AUTOREGRESSION

The use of Bayesian priors on parameter values in a VAR simply involves the incorporation of our prior beliefs about parameter values and their distributions into the estimation technique. The exclusion of variables in a structural model really involves the imposition of our very strong prior beliefs that parameter values on those variables are zero with certainty. Incorporating such strong beliefs about parameter values, even if non-zero, opens the model to criticism of the exclusion of particular variables or of the "known" value of a parameter. Consequently we wish to make broader, more general assumptions about the distributions of parameter values. These priors have sometimes been referred to in the literature as Minnesota or Litterman priors.

A prior distribution is to be created for each estimated parameter. With a quarterly VAR of five variables and six lags, this would involve thirty parameters per equation (assuming no constant) times five equations for a total of 150 prior distributions to specify. Although this can be done, it is by no means an easy task. Also, the individual specification of each parameter's prior could be subject to critical arguments about its specification. Instead, a method to specify more general and accessible priors is developed that alleviates the task of specifying each prior individually, weakens the basis for criticism of individual priors, and allows reproduction by other investigators.

The general specification of the Litterman priors revolves around the assumption that each equation follows a random walk process; $Y_t = Y_{t-1} + v_t$. That is, the expectation of this period's value of the dependent variable is simply last period's observation of that value. The prior mean for the parameter on the first lag of the dependent variable will be one and the prior mean on all other variables will be zero.

Since such naive forecasts are restrictive and unsatisfactory for a variety of reasons, prior distributions are placed around these means. One assumption of these distributions is that variable lags further into the past have less explanatory power than more recent lags. The resulting distributions are illustrated in Figure 1.

Distributions around the prior means must still be quantified. These distributions are specified with general priors imposed in the form of standard deviations of the estimated parameters. The Litterman prior on standard deviations is of the form[2]

$$\delta_{ij}^{\ell} = \begin{cases} \dfrac{\lambda}{\ell^{\gamma_1}} & \text{if } i = j \\[2ex] \dfrac{\lambda \, \gamma_2 \, \hat{\sigma}_i}{\ell^{\gamma_1} \, \hat{\sigma}_j} & \text{if } i \neq j \end{cases}$$

where δ_{ij}^{ℓ} is the standard deviation of the coefficient lag ℓ of variable j in equation i.

---

[2] See Litterman [5], and Bessler and Kling [1]. Note that the RATS manual [Doan and Litterman] is incorrect in its specification of this prior. The subscripts on the scale factors, σ_i and σ_j, are reversed.

The investigator must specify three parameters (λ, γ_1, γ_2) to derive the standard deviations. λ is the constant overall tightness of the prior.[3] As this parameter is set very tightly (approaches zero), the estimated coefficients in the model approach their prior means since the distributions around these means become spiked.

γ_1 is a decay parameter which determines the rate at which lags farther back receive less weight (become tighter around their means). Bessler and Kling use a harmonic lag decay of the form $g(\ell) = \ell^{-\gamma_1}$. They note, however, that $\gamma_1 = 0$ gives better results than other values of γ_1. This means that there is no decay, and that past lags receive weights equal to more recent lags. This structure would not look like Figure 1. Instead all lags would have distributions similar to lag 1 around their respective means.

Finally, γ_2 is used to specify the relative weights of variables in each equation. Own lags of dependent variables typically carry a weight of 1.0. Other variables would be assigned weights ranging from 0.0 to 1.0. Other variables can receive the same weight in each equation or the weights can be more finely tuned to each individual

---

[3] λ is described by Litterman and Bessler and Kling as the "constant standard deviation on the first lag of the dependent variable in each equation." For higher lags, λ is then adjusted by the decay term and the appropriate weights, $\gamma 2$. This interpretation is correct in the true form of the Litterman prior. However, as the prior structure is altered so that the random walk assumption is changed (as will be done in section III) λ should be reinterpreted as a general prior standard deviation for all lags which is modified for individual sets of parameters.

FIGURE 1

A Schematic Representation of the Prior *



* Source: Litterman [6]

equation (Bessler and Kling [1]).[4]

Before estimation, priors are scaled by a ratio of standard errors of univariate autoregressions of the same lag length to be estimated in the VAR. $\hat{\sigma}_i$ is the standard error of residuals from such an autoregression for variable i. This scaling is to ensure that individual variables do not receive inappropriate weights in their contributions to the VAR merely because of the magnitude of their units of measurement.

Once the prior means and standard deviations of the parameters are specified, the equation coefficients can be estimated using a form of Theil mixed estimation (Theil [8]). This is a method of incorporating prior information about the equation parameters into the estimation procedure.

Let $y = X\beta + u$ be the general linear statistical model, and let our series of "dummy observations" on the parameter values be described as $r = R\beta + v$. ($var(u) = var(v) = \sigma^2$.) The Theil mixed estimator is

$$\beta_m = (X'X + R'R)^{-1} (X'y + R'r).$$

Given the prior on an individual parameter as $\beta_i \sim N(b_i, \Theta_i^2)$, a restriction on an individual βi is $r_i = R_i \beta i$ where $R_i = \sigma/\Theta$ and $r_i = (\sigma/\Theta_i) b_i$. b is the prior mean. For a set of these restrictions, in matrix notation, R becomes a diagonal matrix with σ/Θ_i in the ii^{th} entries along the diagonal. With mean 1 on the first lag of the

---

4 See the RATS manual [Doan and Litterman] for "symmetric" and "general" prior specifications.

dependent variable, r becomes a vector with σ/Θ as the entry in the $(\ell*(i-1)+1)^{th}$ cell and zeros everywhere else, where this would be the entry for the first lag of the dependent variable in equation i.[5]

Since σ^2 is unknown, we substitute s^2, where s^2 is the variance of the residuals from a univariate autoregression of the same lag length as the VAR. We have specified our Θ's previously using the notation δ_{ij}^{ℓ}. The entries in the R and r arrays then become s/δ_{ij}^{ℓ}. This is a slightly different approach taken from that appearing in Litterman [5] or Bessler and Kling [1].[6] The R and r arrays as specified above are then used to derive the Theil mixed estimation parameter estimates.

One further comment is necessary to clarify the degrees of freedom in VAR's. Since dummy observations are added for lags of each system variable, the number of observations increases. The degrees of freedom are no longer T-K, where T is the number of observations and K is the number of regressors. Instead they are T-D, where D is the number of deterministic or non-system variables.

---

[5] This notation is for single equation estimation. The arrays R, r, and b would expand if we thought of this procedure as estimating the system simultaneously.

[6] However, it is consistent with the RATS manual and the way RATS computes the priors. Bessler and Kling have misspecified the array r; in their paper. r is not simply the vector of means (ones and zeros). That vector is b. With the use of Litterman priors, the calculation of the Theil mixed estimators remains the same. However, if the priors are generalized to account for seasonality, more care is required in the specification of the R and r arrays.

III. MODELING THE U.S. TURKEY MARKET WITH SEASONAL PRIORS

The U.S. turkey market has an important seasonal component. Traditionally, turkey production was primarily for the holiday season in November and December. However, changes in consumer demand for turkey meat over the past two decades have made turkey production relatively less seasonal.

Because of the seasonality in the market, the use of true Litterman priors may be inappropriate. This section of the paper discusses the turkey market in slightly more detail, presents results of a Bayesian VAR of the turkey market using Litterman priors, and finally compares that model to one which alters those priors to reflect the seasonality in the market.

A. The U.S. Turkey Market

The U.S. turkey industry has evolved from a highly seasonal industry geared primarily for the traditional holiday season to one that produces year round. This is primarily due to shifts in consumer preferences away from red meat consumption. Aggressive marketing of turkey products has also contributed to this shift in preferences. Non-seasonal consumption is also supported by the wide variety of turkey products now available. In 1980, 39% of U.S. turkey production was marketed as processed products (hot dogs, ham, sandwich meat, etc.) 28% as cut up parts, 28% as whole processed birds (smoked, basted, etc.) and only 5% as plain whole birds. Table 1 illustrates shifts in consumer demand for turkey since 1960.

Table 1: Per Capita Turkey Consumption by Quarter[1]

Year	Q1	Q2	Q3	Q4	Total
	- - - - - - - - - - - - -		pounds	- - - - - - - - - -	
1960	.6	.8	1.3	3.4	6.1
1965	.7	.8	1.8	4.1	7.4
1970	.9	.9	2.2	4.2	8.2
1975	1.1	1.4	2.0	4.0	8.5
1980	1.8	2.0	2.7	4.0	10.5
1983	2.1	2.2	2.5	4.4	11.2

[1] Source: Lasley, Henson, and Jones.

Changes in relative meat prices have also contributed to this shift in demand. Table 2 shows how turkey prices have fallen relative to pork and beef prices while roughly remaining on par with chicken prices.

Table 2: Amounts of Other Meats Equal to the Cost of One Pound of Turkey Meat at Retail[1]

Years	Pork	Beef	Chicken
	- - - - - - - - - - -	pounds	- - - - - - - - - -
1960-64	.86	.63	1.22
1965-69	.70	.57	1.21
1970-74	.70	.51	1.31
1975-79	.59	.51	1.31
1980-82	.59	.39	1.28

[1] Source: Lasley, Henson, and Jones.

Note, however, that fourth quarter per capita turkey consumption is still about twice that of each of the first two quarters. To meet this seasonal demand, production still takes place in late summer and early fall, and cold storage inventories are built up prior to the holiday season. Table 3 illustrates 1983 quarterly slaughter and hatch relationships and Table 4 shows cold storage inventories. There is also some seasonality in turkey price although it is not as pronounced. This is shown in Table 5.

Table 3: Quarterly Percentages of 1983 U.S. Turkey Hatch and Slaughter[1]

Quarter	Slaughter %	Hatch %
I	18.0	26.29
II	22.6	33.88
III	29.7	21.91
IV	29.7	17.92

Total Poults Hatched = 182,122,000
Total Turkey Production = 2,563 million pounds RTC

[1] Source: Lasley, Henson and Jones.

Table 4: U.S. Beginning of Quarter Turkey Stocks[1]

	Year			
Quarter	1970	1975	1980	1985
	- - - - - - million pounds RTC - - - - - - -			
I	191.9	272.0	240.0	203.9
II	101.1	207.3	208.9	185.3
III	94.7	193.2	286.6	255.7
IV	343.0	409.8	398.8	432.2

[1] Source: Lasley, Henson and Jones.

Table 5: Average Quarterly Turkey Price[1]

Quarter	1980	1981	1982	1983	1984	1985
	Year					
	- - - - - - cents per pound, 1984 dollars - - - - - -					
I	77.6	72.5	60.7	58.3	68.7	67.5
II	68.0	73.4	63.7	60.0	67.2	63.1
III	85.1	70.5	69.5	62.4	71.9	74.9
IV	88.6	61.1	67.5	71.2	89.3	85.9

[1] Source: Lasley, Henson and Jones.

B. A Bayesian VAR of the U.S. Turkey Market with Litterman Priors

A quarterly Bayesian VAR of the U.S. Turkey Market was estimated with five system variables of six lags each. The system variables are turkey production measured in million pounds RTC (Ready To Cook), the New York wholesale turkey price for 8-16 pound hens measured in dollars per pound, beginning of quarter cold storage inventory in million pounds RTC, turkey poult hatch levels in 1,000 poults, and per capita disposable income. In addition, non-system "exogenous" variables were included. These are two lags each of wholesale beef and chicken prices, six lags each of corn and soybean meal prices, a constant term, a trend term, a dummy term for late 1972 and 1973, and seasonal dummy variables. All prices are in 1984 dollars inflated by the CPI, and income is also in 1984 dollars adjusted by the PCE price deflator.

Input and substitute commodity prices are included as exogenous to the system because the turkey market does not produce enough information to adequately model these prices. These prices would be better modeled with levels of these commodities included as well to generate capable forecasting equations for these series. It is important to have good equations for these price series since forecasts of these prices are used in generating forecasts of the turkey market variables through the chain rule of forecasting. These variables are treated exogenously to keep this model as small as possible while still maintaining the influence of these exogenous commodity prices. As the model stands now, actual historic values of chicken, beef, corn and soybean meal prices are fed into the within sample forecasts for the turkey market instead of forecasts of these exogenous prices.

The prior means placed on the estimated parameters are one for the first lag of the dependent variable in each equation and zero elsewhere. A relatively loose overall tightness of $\lambda = .2$ was used with this model. The decay parameter is $\gamma_1 = 1.0$. These values of λ and γ_1 correspond to those used in the model adjusted for seasonality. These values were chosen after some experimentation with the model and their choice was based on the forecasting ability of the model. A loose value or λ seems reasonable in that it allows for greater deviation from the previous period's (year's) value of the dependent variable resulting from changes in the other variables.

Finally, the weights of other variables in each equation are given in Table 6. These weights were chosen in a rather ad hoc manner based on prior knowledge of the market. Note that these weights are specified so that per capita income is not really influenced by the turkey market and is essentially a six lag

Table 6. General Specification of the Prior, γ_2.

Dependent Variable	Independent Variable Lags				
	PROD	PRICE	STORE	HATCH	PCDI
PROD	1.	.8	.8	.9	.1
PRICE	.9	1.	.9	.8	.1
STORE	.9	.9	1.	.8	.1
HATCH	.8	.9	.8	1.	.1
PCDI	.001	.001	.001	.001	1.

PROD = turkey production, PRICE = turkey price, STORE = turkey stocks, HATCH = turkey hatch, PCDI = per capita disposable income.

univariate autoregression. No priors are placed on the deterministic variables.

The seasonality inherent in the turkey market suggests that the decay factor, γ_1, should not be very large. A large decay would drive the coefficient on the fourth lag of the dependent variable toward zero. We would expect, however, that this coefficient would be close to one. This same model was estimated with $\gamma_1 = 0.0$. This specifies there is no decay on the lag structure.

Table 7 compares these two models. Note that when $\gamma_1 = 0.0$ the equations seem to fit the data much more closely based on R^2 and standard error (SEE) criteria. However, $\gamma_1 = 1.0$ seems to do a much better job of forecasting as shown by the root mean squared forecast errors for the various step-ahead forecasts listed. Perhaps, then, the first equation is overfit to the data. Although it will be shown later that the priors placed on the model with $\gamma_1 = 1.0$ are intuitively wrong, it is important to understand that through the manipulation of, and the experimentation with the values of the parameters that make up the priors, it is very easy to increase the

Table 7: Comparison of the Bayesian VAR's With Different Decay Structures

	$\gamma_1 = 1.0$			$\gamma_1 = 0.0$		
	R^2	SEE*	DW	R_2	SEE	DW
PROD	.9976	13.14	1.77	.9990	8.53	1.63
PRICE	.9785	.0392	2.29	.9885	.0287	2.85
STORE	.9947	10.48	2.22	.9974	7.32	2.69
HATCH	.9975	995.0	2.40	.9986	728.5	2.38
PCDI	.9908	21.74	2.02	.9910	21.43	2.14

Root Mean Squared Forecast Error:

EQUATION	STEP	$\gamma_1 = 1.0$	$\gamma_1 = 0.0$
PROD	1	40.42	52.40
	2	48.19	72.23
	4	61.89	92.38
	8	71.20	143.84
PRICE	1	.151	.182
	2	.214	.280
	4	.208	.298
	8	.224	.599
STORE	1	37.80	48.28
	2	45.75	59.38
	4	51.31	100.50
	8	91.16	156.91
HATCH	1	4480	5475
	2	4600	6139
	4	4702	7138
	8	8793	11501
PCDI	1	36.28	39.40
	2	51.46	55.75
	4	60.98	66.64
	8	73.40	69.48

* SEE = Standard Error of Estimate
DW = Durbin-Watson Test Statistic

R^2 and reduce the standard error. The resulting estimated equations, however, generally will not give accurate forecasts.

C. A Bayesian VAR of the U.S. Turkey Market with Seasonal Priors

The seasonality in the turkey market raises serious concerns about the assumption of the random walk process underlying the VAR methodology. Rather than specify the random walk process as $Y_t = Y_{t-1} + v_t$ in a quarterly model, a more appealing specification would be $Y_t = Y_{t-4} + v_t$. Under this assumption the prior mean would be one on the fourth lag of the dependent variable in each equation and zero elsewhere. Distributions around the means must also change so that variance decays immediately on the first lag but remains at full value on the fourth lag. The resulting distribution would look like that in Figure 2.

The same model described in B. above was estimated with seasonal priors. Means on lags of the dependent variables are one on the fourth lag and zero everywhere else. The per capita disposable income variable retains the Litterman prior structure, however. The decay structure is:

$$g(\ell) = \begin{cases} (\ell+1)^{\gamma_1} & \ell = 1,2,3 \\ \ell^{\gamma_1} & \ell = 4 \\ (\ell-3)^{\gamma_1} & \ell = 5,6 \end{cases}$$

where $\gamma_1 = 1.0$. Everything else in the model remains as it appears for the previous VAR's.

FIGURE 2

A Schematic Representation of the Prior *



* Format from Litterman [6]

Table 8: A Comparison of Seasonal and Litterman Priors in VAR's of the U.S. Turkey Market

EQUATION	PROD	PRICE	STORE	HATCH	PCDI
R^2					
SBVAR	.995	.943	.986	.995	.990
BVAR	.999	.989	.997	.999	.991
SEE					
SBVAR	19.18	.064	17.2	1439.7	23.08
BVAR	13.14	.039	10.48	995.0	21.74

		RMSE		THEIL U	
EQUATION	STEP	SBVAR	BVAR	SBVAR	BVAR
PROD	1	13.14	40.42	.058	.179
	2	36.72	48.19	.124	.162
	4	45.50	61.89	.994	1.352
	8	61.29	71.20	1.109	1.289
PRICE	1	.040	.151	.422	1.579
	2	.112	.214	.868	1.665
	4	.120	.208	.930	1.609
	8	.138	.224	.759	1.231
STORE	1	12.87	37.80	.085	.249
	2	34.38	45.75	.186	.247
	4	39.71	51.31	.747	.965
	8	47.99	91.16	.7719	1.464
HATCH	1	941	4480	.057	.272
	2	2598	4600	.114	.202
	4	2448	4702	.906	1.741
	8	4027	8793	1.142	2.494
PCDI	1	18.38	36.28	.645	1.274
	2	37.24	51.46	.793	1.096
	4	55.77	60.98	.696	.739
	8	77.38	73.40	.537	.510

Table 8 compares the VAR with seasonal priors (SBVAR) to the model with Litterman priors (BVAR). Note that the model with seasonal priors outperforms the other model based on the forecast statistics RMSE and THEIL U. The THEIL U statistic is greater than one only twice for the steps listed.[7] This statistic has little meaning for steps 1 and 2, however, because of the seasonality of the data. It can be useful for the comparison of two or more models, but the low values for these steps is to be expected since the model hopefully does a better job accounting for seasonality than the naive forecast of no change.

These forecast statistics are for out of sample forecasts from within the sample data. The model is first estimated over a subperiod of the data; in this case from third quarter 1971 to first quarter 1979. Forecasts are then made and compared to the data not yet included in the estimation. One more observation is then added and the model is reestimated using the Kalman filter. Forecast statistics are again computed. This process is continued until

---

[7] The Theil U statistic is the ratio of the RMSE of the model to the RMSE of the naive forecast of no change. It is specified for a k-step ahead forecast as:

$$U = + \left(\frac{\sum_t (F_{t+k} - A_{t+k})^2}{\sum_t (A_t - A_{t+k})^2} \right)^{1/2}$$

where F_t = forecast value of the dependent variable

A_t = actual value of the dependent variable.

This statistic is convenient because it is independent of units of measure. Values of $U > 1.0$ indicate the naive forecast outperforms the model. They are not very impressive.

fourth quarter 1984. This process, then, generates forecast statistics which evaluate the specification of the model in terms of variables, lags, etc., instead of evaluating final parameter values.

The improvement in the forecast statistics shown in the SBVAR model is quite dramatic. The most improvement is in the PRICE equation. The model with seasonal priors forecasts price reasonably well, while the model with Litterman priors forecasts PRICE very poorly.

IV. SUMMARY AND CONCLUSIONS

This paper has presented a brief discussion of the use of Bayesian priors in vector autoregression models. Three models of the U.S. turkey market are presented. The first two used standard Bayesian VAR techniques and differed only in decay structure. The last model altered the standard use of the Litterman priors to account for seasonality in the market. That procedure dramatically improved the forecasting performance of the model.

The results discussed in the previous section indicate that the intelligent use of Bayesian priors in a manner appropriate to the problem at hand can yield significant improvement in the model results. If one remembers that the Bayesian priors represent what we know about the model a priori, then the priors can be tailored to a wide variety of problems. This method has the flexibility to be useful in a great many applications.

More work needs to be done in this area in a number of places. A standard method of putting priors on the deterministic variables is needed. The method to deal with seasonality presented in this paper needs to be refined. The method to determine the relative weights of variables (γ_2) in the equations needs more work. Perhaps these needs can be met as VAR's are applied to more economic problems.

REFERENCES

1. Bessler, D., and J. Kling, "Forecasting Vector Autoregressions with Bayesian Priors," American Journal of Agricultural Economics, Vol. 68, Number 1, February 1986.

2. Doan, T. and R. Litterman, User's Manual: Regression Analysis of Time Series, VAR Econometrics, 1984.

3. Ford, S., "A Beginner's Guide to Vector Autoregression," Staff Paper P86-28, Department of Agricultural and Applied Economics, University of Minnesota, September 1986.

4. Lasley, F., W. Henson, and H. Jones, "The U.S. Turkey Industry," Agricultural Economic Report Number 525, ERS, USDA, March 1985.

5. Litterman, R., "Techniques of Forecasting Using Vector Autoregressions," Working Paper #115, Federal Reserve Bank of Minneapolis, 1979.

6. Litterman, R., "Specifying Vector Autoregressions for Macroeconomic Forecasting," Research Department, Staff Report 92, Federal Reserve Bank of Minneapolis, March 1984.

7. Sims, C., "Macroeconomics and Reality," Econometrica, Vol 48, No. 1, January 1980.

8. Theil, H., "On the use of Incomplete Prior Information in Regression Analysis," Journal of the American Statistical Association, Vol. 58, June 1963.

9. Todd, R., "Improving Economic Forecasting with Bayesian Vector Autoregression," Quarterly Review, Federal Reserve Bank of Minneapolis, Fall 1984.

APPENDIX

List of Variable Definitions:

δ_{ij}^{ℓ}	=	the prior standard deviation of the estimated parameter of lag ℓ of variable j in equation i (referred to also as Θ_i)
λ	=	the lag length of a particular variable
λ	=	overall tightness parameter in the prior standard deviation
g_1	=	the decay parameter in the prior standard deviation
γ_2	=	weights of lags of the non-dependent variables in prior standard deviation
σ	=	standard deviation of residuals
β	=	estimated parameter vector
r,R	=	restrictions on β
b	=	vector of prior means
S	=	estimate of true residuals in the model (σ)
T	=	number of observations over time in the data set
K	=	number of regressors in an estimated equation
D	=	number of deterministic variables in the VAR system (those not having estimated equations)

```
** THIS PROGRAM IS TO CALCULATE THE THEIL MIXED
** ESTIMATION PARAMETERS FOR A BAYESIAN VAR AND
** COMPUTE THE APPROPRIATE FORECAST STATISTICS
**
** THIS PROGRAM IS WRITTEN FOR MAINFRAME RATS
**
** CHANGES MUST BE MADE ONLY IN LINES MARKED WITH
** @@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
**
** THIS SECTION OF CODE SETS UP THE RATS WORK SPACE
** AND READS IN THE DATA
**
CAL 70 1 4
ALLOCATE 40 88,1 10 150
EQV 1 TO 27
USDA XPORT STORE HATCH SMEAL CORN PROD DIS PRICE $
CHICK PORK BEEF PCDI WTPRC Q2 TREND DUM73 $
E1 E2 E3 E4 E5 F1 F2 F3 F4 F5
CLEAR
DATA(ORG=VAR) 70,1 85,4 1 TO 14
LIMITS ERRORS 100 WARNINGS 1000
**
** THIS SECTION OF CODE DEFINES THE DETERMINISTIC
** VARIABLES TO BE INCLUDED IN THE VAR
**
SEASONAL Q2 70,1 85,4 4 70,2
SET TREND 70,1 85,4 = T
ZEROS DUM73 70,1 72,2
UNITS DUM73 72,3 73,4
ZEROS DUM73 74,1 85,4
** THIS SECTION OF CODE READS IN THE FOLLOWING
** PARAMETERS:
**
**          DETER = NUMBER OF DETERMINISTIC VARIABLES
**          LG = NUMBER OF LAGS
**          NUMV = NUMBER OF SYSTEM VARIABLES
**          DK = DECAY PARAMETER
**          TGHT = TIGHTNESS PARAMETER
**
DECLARE INTEGER DETER LG NUMV
DECLARE REAL DK TGHT
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
INPUT DETER LG NUMV DK TGHT
22 6 5 1. .2
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
**
** THIS SECTION OF CODE PREPARES THE VECTORS
** OF DECAY PRIORS GIVEN DK
**
```

```
**          VDK = VECTOR OF DECAYS BY LAG
**
DECLARE VECTOR VDK(LG)
DECLARE VECTOR VDKB(LG)
**
**          ROWS = NUMBER OF ROWS IN PRIOR MATRIX
**
 IEVAL ROWS=NUMV*LG+DETER+1
 DO I = 1,3
   EVAL VDK(I)=(I+1)**(-DK)
 END
 DO I = 4,LG
   EVAL VDK(I)=(I-3)**(-DK)
 END
 DO I = 1,LG
   EVAL VDKB(I)=I**(-DK)
 END
**
** THIS SECTION OF CODE INPUTS THE ARRAY OF WEIGHTS
** TO BE USED IN THE PRIORS
**
**            PGEN = MATRIX OF GENERAL PRIORS
**
DECLARE RECTANGULAR PGEN(NUMV,NUMV)
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
INPUT PGEN
1. .9 .9 .8 .001 .8 1. .9 .9 .001 .8 .9 1. .8 .001 $
.9 .8 .8 1. .001 .1 .1 .1 .1 1.
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
**
** THIS SECTION OF CODE COMPUTES THE UNSCALED PRIORS
**
**
**            PRIR = INPUT MATRIX OF NON-SCALED PRIORS
**
DECLARE RECTANGULAR PRIR(ROWS,NUMV)
EWISE PRIR(I,J)=0.0
 DO I=1,NUMV
   DO J=1,NUMV
     DO K=1,LG
      IF I.EQ.5
       EVAL PRIR(((J-1)*LG+K),I)=1/(TGHT*VDKB(K)*PGEN(J,I))
      ELSE
      EVAL PRIR(((J-1)*LG+K),I)=1/(TGHT*VDK(K)*PGEN(J,I))
    END
  END
END
**
** THIS SECTION OF CODE DEFINES THE VAR SYSTEM WITH ALL
** PRIOR PARAMETERS SET EQUAL TO ONE.  THIS WAY RATS
** WILL DO THE SCALING.  THIS METHOD IS USED BECAUSE OF
** A BUG IN THE FULL OPTION OF SPECIFY IN MAINFRAME RATS.
```

```
**
**
SYSTEM 1 TO NUMV
**
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
VARIABLES PROD PRICE STORE HATCH PCDI
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
**
LAGS 1 TO LG
**
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
DET CONSTANT -CHICK 1 2 -BEEF 1 2 -CORN 1 6 -SMEAL 1 6 -Q2 0 2 $
TREND DUM73
**@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
**
SPECIFY(TIGHT=1.,DECAY=0.0,TYPE=SYMMETRIC) 1.0
END(SYSTEM)
**
**
** THIS SECTION OF CODE IS TO PERFORM THE ESTIMATION
** AND CALCULATE THE FORECAST STATISTICS
**
**
**          SCPRIR = OUTPUT MATRIX OF SCALED PRIORS
**
DECLARE RECTANGULAR SCPRIR(ROWS,NUMV)
DECLARE RECTANGULAR SCPRIR2(ROWS,NUMV)
IEVAL NREG=ROWS-1
DECLARE REAL PTEMP
DECLARE VECTOR SMALLR(NREG)
DECLARE VECTOR SMALLR2(NREG)
DECLARE SYMMETRIC CMOMXX(NREG,NREG)
DECLARE VECTOR CMOMXY(NREG)
DECLARE RECTANGULAR LPRIR(NREG,NREG)
**
**
** THE ESTIMATION OF THE MODEL IS PLACED INSIDE THE
** THEIL LOOP
**
**
THEIL(SETUP) 5 20 85,4
#1 TO 5
DO DATE=(79,1),(84,4)
**
** THE ESTIMATE IS PLACED HERE TO ENABLE RATS TO SCALE
** THE PRIORS PROPERLY WITH EACH ADDED OBSERVATION
**
**
ESTIMATE(DUMMY=SCPRIR,NOPRINT,NOFTESTS) 71,3 DATE
EVAL PTEMP=0.0
**
** THE PRIORS ARE CALCULATED HERE
```

```
**
DO I = 1,(NUMV*LG)
  DO J = 1,NUMV
    EVAL PTEMP=SCPRIR(I,J)*PRIR(I,J)
    EVAL SCPRIR2(I,J)=PTEMP
  END
END
**
**           SMALLR = VECTOR OF PRIOR MEANS
**
**
**
**          CMOMXX = NEW X'X PORTION OF CMOM
**          CMOMXY = NEW X'Y PORTION OF CMOM
**          LPRIR = TRANSFORMATION OF SCALED PRIORS
**                  TO USABLE DIAGONAL PRIOR MATRIX
**          SMALLR2 = SCALED PRIOR MEAN VECTOR
**
**
** THIS IS THE ACTUAL THEIL MIXED ESTIMATION ROUTINE
**
**
DO K=1,NUMV
  EWISE LPRIR(I,J)=0.0
  EWISE SMALLR(I)=0.0
  IF K.NE.5
  EVAL SMALLR((K-1)*LG+4)=1.0
  ELSE
  EVAL SMALLR((K-1)*LG+1)=1.0
  DO J=1,NREG
    EVAL LPRIR(J,J)=SCPRIR2(J,K)
  END
  MAT SMALLR2=LPRIR*SMALLR
**
**
** THE ESTIMATION TAKES PLACE IN TWO STAGES.  THE MOMENT MATRIX
** IS FORMED FROM THE DATA.  IT IS THEN ALTERED ACCORDING TO
** THE THEIL MIXED ESTIMATION PROCEDURE AND THE REGRESSION IS
** COMPLETED.
**
**
  CMOMENT(EQUATION=K) 71,3 DATE
  OVERLAY CMOM(1,1) WITH CMOMXX(NREG,NREG)
  OVERLAY CMOM(NREG+1,1) WITH CMOMXY(NREG)
  MAT CMOMXX=CMOMXX+TR(LPRIR)*LPRIR
  MAT CMOMXY=CMOMXY+TR(LPRIR)*SMALLR2
  REGRESS(EQUATION=K,DFC=-NUMV*LG,NOPRINT)
END
THEIL DATE
END
THEIL(DUMP)
**
```

```
** THE THEIL MIXED ESTIMATION IS REPEATED AGAIN HERE
** FOR THE WHOLE SAMPLE TO ENABLE THE PRINTING OF THE
** ESTIMATION RESULTS FOR ONLY THE FINAL UPDATE. THIS IS
** AN INCONVENIENCE OF RATS.  ANY SUGGESTIONS REGARDING
** ALTERNATIVES ARE WELCOME.
**
**
DO K=1,NUMV
EWISE LPRIR(I,J)=0.0
EWISE SMALLR(I)=0.0
IF K.NE.5
EVAL SMALLR((K-1)*LG+4)=1.0
ELSE
EVAL SMALLR((K-1)*LG+1)=1.0
DO J=1,NREG
EVAL LPRIR(J,J)=SCPRIR2(J,K)
END
MAT SMALLR2=LPRIR*SMALLR
CMOMENT(EQUATION=K) 71,3 84,4
OVERLAY CMOM(1,1) WITH CMOMXX(NREG,NREG)
OVERLAY CMOM(NREG+1,1) WITH CMOMXY(NREG)
MAT CMOMXX=CMOMXX+TR(LPRIR)*LPRIR
MAT CMOMXY=CMOMXY+TR(LPRIR)*SMALLR2
REGRESS(EQUATION=K,DFC=-NUMV*LG)
END
**
END
```